POGLAVJE II

2. Reševanje diferencialne enačbe nihanja

1. Uporaba integriranja za reševanje diferencialne enačbe nihanja

1. Uporaba integriranja za reševanje diferencialne enačbe nihanja

Čeprav je mogoče s programskim paketom *Mathematica* reševati diferencialne enačbe s posebnim ukazom (ki je bil že prikazan in ne preveč uspešno uporabljen v prejšnjem podpoglavju, ter bo še posebej prikazan v naslednjem podpoglavju), je nekatere enostavne diferencialne enačbe mogoče reševati tudi z integriranjem.
Zgled bo tako, predvsem zaradi demonstracije možnosti integriranja, izveden na primeru reševanja diferencialne enačbe nedušenega lastnega nihanja linearnega sistema, ki je matematično enake oblike kot prva izmed dveh navadnih diferencialnih enačb, ki nastaneta po ločitvi funkcij, katerih produkt predstavlja rešitev parcialne diferencialne enačbe (poglavje II.1). Taka diferencialna enačba pa ni vezana zgolj na iskanje prečnega pomika, temveč ima mnogo bolj splošno uporabnost.
Ker so za rešitev diferencialnih enačb potrebne neke informacije o stanju sistema ob nekih časih, predpostavimo, da gre za sistem, ki se na začetku premakne iz ravnotežne lege za nek pomik, npr. u_o, ter nato spusti, da iz mirovanja ($\dot{u}_o = 0$) prosto niha. Diferencialna enačba takšnega sistema se zapiše (ne glede na začetne oz. robne pogoje) kot

$$m \cdot \ddot{u} + k \cdot u = 0$$

kjer sta m in k masa in togost sistema, $u \equiv u(t)$ pa iskani pomik, ki je seveda funkcija časa. Zveza povezuje pospešek in pomik sistema ter se lahko zapiše tudi kot:

$$m \cdot \ddot{u} = -k \cdot u$$

$$\ddot{u} = -\frac{k}{m} \cdot u = -\omega^2 \cdot u$$

ki direktno podaja pospešek kot funkcijo položaja (upošteva se znana zveza $\omega^2 = \frac{k}{m}$). Ker pa je pospešek odvod hitrosti po času, sledi zveza:

$$\frac{d\dot{u}}{dt} = -\omega^2 \cdot u$$

Nato se uporabi pravilo zaporednega odvajanja, kar vodi do:

$$\frac{d\dot{u}}{dt} \cdot \frac{du}{du} = \frac{d\dot{u}}{du} \cdot \frac{du}{dt} = \frac{d\dot{u}}{du} \cdot \dot{u} = -\omega^2 \cdot u$$

Ločitev spremenljivk vodi do integrabilne zveze (ob vpeljavi zapisa $\dot{u} = v$):

$$\dot{u} \cdot d\dot{u} = -\omega^2 \cdot u \cdot du$$

$$v \cdot dv = -\omega^2 \cdot u \cdot du$$

Integracija leve strani tako vodi do (ker bo z rezultatom še potrebno operirati, je smiselno, da se ga shrani v neko spremenljivko):

```
In[1]:= leva = Integrate[v, v]
```

$$\text{Out[1]= } \frac{v^2}{2}$$

Integrirati pa je potrebno še desno stran in upoštevati, da *Mathematica* pri integraciji ne zapiše integracijske konstante avtomatično. Zato jo je potrebno dodati posebej, npr. C_1 (pri tem je popolnoma vseeno, na kateri strani enačbe se pripiše):

```
In[2]:= desna = Integrate[-ω^2 u, u] + C1
```

$$\text{Out[2]= } C1 - \frac{u^2 \omega^2}{2}$$

Izenačitev leve in desne strani sedaj predstavlja zvezo, ki poveže hitrost in položaj sistema:

$$\frac{v^2}{2} = C_1 - \frac{u^2 \cdot \omega^2}{2}$$

Če se želi integracijska konstanta določiti sproti oz. na tem nivoju analize (ko je torej podana zveza med pomikom in hitrostjo), je tako potrebno poznati informacijo o neki hitrosti ob pripadajočem položaju (druga možnost je seveda pridobitev končne informacije o pomiku u(t) ter šele nato vstavitev vseh informacij ter določitev vseh integracijskih konstant), kar je v obravnavanem primeru znano. Tako kot je to tudi najbolj običajno, je ta informacija namreč sedaj znana na začetku gibanja. Znano je namreč, da se sistem na začetku (t=0 s) premakne za nek odmik, npr. u_0 iz ravnotežne lege, nato pa se sistem iz mirovanja spusti (v=0), da prosto niha. Tako je konstanto mogoče izračunati iz (ene) enačbe z (eno) neznanko z ukazom *Solve[enačba, neznanka]*, čeprav je ukaz sposoben reševati tudi sistem enačb:

```
In[3]:= Solve[leva == desna /. {u → u0, v → 0}, C1]
```

$$\text{Out[3]= } \left\{\left\{C1 \to \frac{u0^2 \omega^2}{2}\right\}\right\}$$

Z opcijo */.{u→u₀, v→v₀}* je bilo programu ukazano, naj zgolj pri izenačitvi leve in desne strani upošteva, da je $u=u_o$ ter $v=v_o$ (torej robna pogoja). Čeprav je vrednost konstante sedaj znana, njena vstavitev v enačbo ne prinese nikakršnega napredka (saj gre zgolj za konstanto), ampak zgolj razširi zapis rešitve (kar pa lahko posledično *Mathematici* povzroči dodatne ovire). Ker je (zaradi enostavnosti izračuna) smiselno ohranjati čim kompaktnejši zapis, se vrednost izračunane konstante zato shrani v neko spremenljivko (katere ime se lahko izbere dovolj asociativno), njena vstavitev v zvezo pa se zgolj začasno odloži:

```
In[4]:= c1 = C1 /. %[[1]]
```

$$\text{Out[4]= } \frac{u0^2 \omega^2}{2}$$

Ukaz v spremenljivki c_1 (ki seveda ni ista kot C_1, ki še naprej ostaja nedefinirana) shrani vrednost za konstanto C_1, izračunano v prejšnji vrstici. Če pa bi se zapisalo *c_1=%,* bi *Mathematica* ne shranila zgolj vrednosti rešitve, ampak celoten izraz rešitve, torej z oklepaji vred.

Iz zveze med hitrostjo in položajem je sedaj (za potrebe integriranja) potrebno eksplicitno izraziti hitrost kot funkcijo položaja:

```
In[5]:= Solve[leva == desna, v]
```

$$\text{Out[5]=}\ \left\{\left\{v\to -\sqrt{2}\ \sqrt{C1-\frac{u^2\,\omega^2}{2}}\right\},\ \left\{v\to \sqrt{2}\ \sqrt{C1-\frac{u^2\,\omega^2}{2}}\right\}\right\}$$

Rezultata je mogoče še nekoliko poenostaviti:

```
In[6]:= Simplify[%]
```

$$\text{Out[6]=}\ \left\{\left\{v\to -\sqrt{2\,C1-u^2\,\omega^2}\right\},\ \left\{v\to \sqrt{2\,C1-u^2\,\omega^2}\right\}\right\}$$

Upošteva se drugi, torej pozitivni izraz in z ukazom se torej kot vrednost spremenljivke upošteva druga rešitev:

```
In[7]:= v = v /. %[[2]]
```

$$\text{Out[7]=}\ \sqrt{2\,C1-u^2\,\omega^2}$$

kamor se sedaj lahko vstavi že znana in izračunana vrednost konstante C_1:

```
In[8]:= Simplify[v /. C1 → c1]
```

$$\text{Out[8]=}\ \sqrt{(-u^2+u0^2)\,\omega^2}$$

Da se izpod korena eliminira vrednost ω^2 (*Mathematica* izrazov pod koreni namreč ne nadomešča avtomatično z vrednostjo njihovega korena), se uporabi ukaz *PowerExpand[izraz]*:

```
In[9]:= v = PowerExpand[%]
```

$$\text{Out[9]=}\ \sqrt{-u^2+u0^2}\ \omega$$

Sedaj dobljena zveza se v standardnem matematičnem zapisu zapiše kot:

$$v = \omega\cdot\sqrt{-u^2+u_o{}^2}$$

Ker pa je hitrost odvod položaja po času, sledi:

$$\frac{du}{dt} = \omega\cdot\sqrt{-u^2+u_o{}^2}$$

in ločitev spremenljivk

$$\frac{du}{\omega\cdot\sqrt{-u^2+u_o{}^2}} = dt$$

vodi do integrabilne zveze. V *Mathematici* se tako zapiše:

In[10]:= **du = FullSimplify[1 / v]**

Out[10]= $\frac{1}{\sqrt{-u^2 + u0^2}\ \omega}$

Ukaz *FullSimplify[izraz]* je podoben oz. v bistvu razširjen ukaz *Simplify[izraz]*, saj za poenostavitev izraza uporabi širšo množico vgrajenih (in uporabniku sicer nevidnih) transformacij, s katerimi poskuša doseči poenostavitev izraza. Integral leve strani je tako:

In[11]:= **Integrate[du, u]**

Out[11]= $-\frac{\text{ArcTan}\left[\frac{u\sqrt{-u^2+u0^2}}{u^2-u0^2}\right]}{\omega}$

ki pa ga je mogoče še poenostaviti:

In[12]:= **Simplify[%]**

Out[12]= $\frac{\text{ArcTan}\left[\frac{u}{\sqrt{-u^2+u0^2}}\right]}{\omega}$

Integral desne strani je mogoče izvesti kar direktno brez uporabe ukaza *Integrate[]* in je enak času *t*. Tako sledi zveza, ki povezuje položaj in čas (na eni strani je potrebno dodati še integracijsko konstanto). Iz nje je mogoče izraziti položaj *u* kot funkcijo časa *t*:

In[13]:= **Solve[t == % + C2, u]**

Out[13]= $\left\{\left\{u \to -\frac{u0\ \text{Tan}[(C2 - t)\ \omega]}{\sqrt{1 + \text{Tan}[(C2 - t)\ \omega]^2}}\right\}, \left\{u \to \frac{u0\ \text{Tan}[(C2 - t)\ \omega]}{\sqrt{1 + \text{Tan}[(C2 - t)\ \omega]^2}}\right\}\right\}$

Rešitvi se najprej poenostavita z ukazom *Simplify[]*:

In[14]:= **Simplify[%]**

Out[14]= $\left\{\left\{u \to -\frac{u0\ \text{Tan}[(C2 - t)\ \omega]}{\sqrt{\text{Sec}[(C2 - t)\ \omega]^2}}\right\}, \left\{u \to \frac{u0\ \text{Tan}[(C2 - t)\ \omega]}{\sqrt{\text{Sec}[(C2 - t)\ \omega]^2}}\right\}\right\}$

še boljši učinek pa se doseže z ukazom *PowerExpand[]*:

In[15]:= **PowerExpand[%]**

Out[15]= $\{\{u \to -u0\ \text{Sin}[(C2 - t)\ \omega]\}, \{u \to u0\ \text{Sin}[(C2 - t)\ \omega]\}\}$

Sedaj je končno mogoče definirati funkcijo, ki podaja položaj kot funkcijo časa:

```
In[16]:= u[t_] = u /. %[[2]]

Out[16]= u0 Sin[(C2 - t) ω]
```

Dokočna rešitev pa bo jasno znana in uporabna šele, ko znana tudi vrednost konstante C_2. Spet se uporabi informacija, da se sistem na začetku (t=0 s) premakne za odmik u_o iz ravnotežne lege, kar vodi do:

```
In[17]:= Solve[u[0] == u0, C2]

         Solve::ifun : Inverse functions are being used
           by Solve, so some solutions may not be found.

Out[17]= {{C2 → π/(2 ω)}}
```

Ker je sinus periodično ponavljajoča se funkcija, *Mathematica* opozori, da najdena rešitev ni nujno tudi edina možna (kar je tudi korektno in resnično).

Vrednost konstante C_2 je tako:

```
In[18]:= C2 = C2 /. %[[1]]

Out[18]= π/(2 ω)
```

kar vodi do končne rešitve problema:

```
In[19]:= u[t_] = Simplify[u[t]]

Out[19]= u0 Cos[t ω]
```

O pravilnosti rešitve se je mogoče najenostavneje prepričati tako, da se rešitev vstavi v začetno informacijo, torej v diferencialno enačbo gibanja:

```
In[20]:= u''[t] + ω^2 u[t]

Out[20]= 0
```

kar očitno potrdi korektnost rešitve.

Zgled je prikazal možnost uporabe integriranja za reševanje diferencialnih enačb. Čeprav je za reševanje takšnih enačb na voljo ustreznejši ukaz, ki privede do enake rešitve na precej elegantnejši način, leži pomen tega zgleda v prikazu možnost uporabe različnih orodij za poenostavitev rezultatov.

NALOGA ZA SAMOSTOJNO DELO

1. Reši stabilitetni problem prostoležečega nosilca (reši diferencialno enačbo in poišči Eulerjevo kritično uklonsko silo).

Diferencialna enačba stabilitetnega problema je:

$$EI \cdot v''(x) + P_{ukl} \cdot v(x) = 0$$

REŠITEV

Namig: pri reševanju uporabi ukaz *ExpToTrig[izraz]*

Kritične sile $P_{ukl,n} = n^2 \cdot \dfrac{EI \cdot \pi^2}{L^2}$

pripadajoče upogibnice

$$v_1(x) = -\frac{L \cdot \sin\left(\frac{\pi \cdot x}{L}\right)}{2 \cdot \pi} \qquad v_2(x) = \frac{L \cdot \sin\left(\frac{2 \cdot \pi \cdot x}{L}\right)}{4 \cdot \pi} \qquad v_3(x) = -\frac{L \cdot \sin\left(\frac{3 \cdot \pi \cdot x}{L}\right)}{6 \cdot \pi}$$

oziroma

$$v_n(x) = (-1)^n \cdot \frac{L \cdot \sin\left(\frac{n \cdot \pi \cdot x}{L}\right)}{2 \cdot n \cdot \pi}$$